

Zeitreihen

Die Prozedur **Zeitreihen** ermittelt längste Serien gleichbleibend guter oder schlechter Ergebnisse.

Es gibt drei Auswertungen **HA(0)**, **HA(1)**, **HA(2)**, die in der Schleife **For j = 0 To 2** durchlaufen werden. Je nach Index **j** wird die Abfrage **Sql** der Spiele modifiziert. **J=0** nur Heimspiele, **j=1** nur Auswärtsspiele, **j=2** alle Spiele (siehe Tabelle auf der vorigen Seite).

Innerhalb jeder dieser Auswertungen wird nach fünf Serien gesucht (**Text(0..4)**). Zur Ermittlung der Serien werden vier Variablen eingesetzt: **SMax** (speichert die Spielnummer der längsten gefundenen Serie), **CMax** (speichert die Anzahl der zu dieser Serie gehörigen Spiele), **S** (die Spielnummer der gerade untersuchten Serie), **C** (die Anzahl der zusammenhängenden Spiele dieser Serie). Wenn das **C** der gerade untersuchten größer ist als die gespeicherte Folge **CMax**, wird **S** zum neuen **SMax** und **C** zum neuen **CMax**.

Die fünf Serien werden in der Schleife **For Serie = 0 To 4** gesucht. Ausgewertet wird das Feld **Sieg**, das die drei Werte **S**, **U** oder **N** haben kann.

Wenn eine Serie noch nicht begonnen hat **C(Serie)=0**, startet sie mit der ersten zutreffenden Übereinstimmung. Ist eine Serie zu Ende **Else**, wird geprüft, ob die aktuelle Serie länger ist als die bereits gespeicherte. Wenn ja, tritt sie an ihre Stelle und der Zähler für die aktuelle Serie wird zurückgesetzt **S(Serie)=Spiel** und **C(Serie)=0**.

Die Ausgabe der gespeicherten Ergebnisse erfolgt durch Konstruktion der **Sql**-Abfrage aus den ermittelten Werten **SMax** und **CMax** kombiniert mit einer Einschränkung für den Spielort **Sql = " (Heim='H') AND "**.

Die Ausgabe selbst erfolgt im Zuge von **Debug.Print**-Anweisungen, die einen einfachen Text ausgeben.

Damit dieser Text in eine Word-Tabelle umgeformt werden kann, wird er noch in Notepad++ mit der auf der vorigen Seite angegebenen Regular Expression umformatiert.

Dateneingabe

Wie füttert man die Datenbank mit neuen Spielern, Vereinen und Spielen? Man könnte natürlich Daten direkt in die Tabellen eingeben. Das sollte man aber nicht, speziell, wenn es redundante Felder gibt, die erst aus den Eingaben errechnet werden. Man gibt daher neue Daten über Formulare ein, die eine fehlerfreie Eingabe durch Plausibilitätschecks auf dem Umweg kleiner Funktionen überprüfen können.

Die Daten der ÖFB-Seite sind bis zum letzten Spiel im Jahr 2012, Spielnummer 716 gegen die Elfenbeinküste komplett. Die Daten bis zum 727 Länderspiel gegen Uruguay müssen händisch nachgetragen werden. Als Quelle dient die Seite weltfussball.at.

Auf der folgenden Seite sieht man die Formulare zur Eingabe von Spielern, Vereinen und Spielen.

Noch bevor die Daten zu einem konkreten Spiel eingegeben werden, prüft man, ob es neue Spieler oder neue Vereine gibt. Diese gibt man zuerst ein.

Um ein Formular zu erstellen, genügt es, die Tabelle zu öffnen und auf *Erstellen->Formular* zu klicken. Abgesehen von Fragen der Gestaltung werden alle erforderlichen Felder im Formular eingesetzt und man kann schon mit der Eingabe beginnen.

Die beiden Eingabemasken (**nächste Seite rechts oben**) zeigen, wie Spieler und Verein zusammenhängen. Man kann zu einem Spieler mehrere Vereine zuordnen und umgekehrt einem Verein mehrere Spieler.

Bei der Eingabe der Mannschaftsaufstellung gibt man daher immer eine Spieler-Verein-Kombination aus der Tabelle **Spieleler-Verein** ein.

Das wichtigste Eingabeformular ist aber **Spiel**:

Es besteht aus dem Hauptformular mit den Angaben zum Spiel im oberen Teil und einem Unterformular für die teilnehmenden Spieler und einem Unterformular für die Tore des Spiels.

```
Public Sub Zeitreihen()
    Dim RS_Spiele As Recordset
    Dim Text(5) As String
    Dim HA(3) As String
    HA(0) = "Serien bei Heimspielen"
    HA(1) = "Serien bei Auswärtsspielen"
    HA(2) = "Serien bei Heim- und Auswärtsspielen"
    Dim SMax(5) As Integer
    Dim CMax(5) As Integer
    Dim S(5) As Integer
    Dim C(5) As Integer
    Dim i, j As Integer
    Text(0) = "Längste Siegesserie"
    Text(1) = "Längste Serie ohne Niederlage"
    Text(2) = "Längste Unentschiedenserie"
    Text(3) = "Längste Serie ohne Sieg"
    Text(4) = "Längste Niederlagenserie"
    Dim Sql As String
    For j = 0 To 2 '0: Heim, 1: Auswärts, 2: Heim und Auswärts
        Debug.Print HA(j)
        For i = 0 To 4
            SMax(i) = 0
            CMax(i) = 0
            S(i) = 0
            C(i) = 0
        Next i
        Select Case j
            Case 0
                Sql = "SELECT Spiel, Sieg FROM Spiel WHERE Heim='H' ORDER BY Spiel"
            Case 1
                Sql = "SELECT Spiel, Sieg FROM Spiel WHERE Heim='A' ORDER BY Spiel"
            Case 2
                Sql = "SELECT Spiel, Sieg FROM Spiel ORDER BY Spiel"
        End Select
        Set RS_Spiele = CurrentDb.OpenRecordset(Sql)
        Do While Not RS_Spiele.EOF
            Dim Sieg, Spiel As String
            Sieg = RS_Spiele("Sieg")
            Spiel = RS_Spiele("Spiel")

            Dim Serie As Integer
            For Serie = 0 To 4
                If (Serie = 0 And Sieg = "S")
                    Or (Serie = 1 And (Sieg = "S" Or Sieg = "U")) _
                    Or (Serie = 2 And (Sieg = "U")) _
                    Or (Serie = 3 And (Sieg = "N" Or Sieg = "U")) _
                    Or (Serie = 4 And (Sieg = "N")) Then
                        If C(Serie) = 0 Then
                            S(Serie) = Spiel
                        End If
                        C(Serie) = C(Serie) + 1
                    Else
                        If C(Serie) > CMax(Serie) Then
                            SMax(Serie) = S(Serie)
                            CMax(Serie) = C(Serie)
                        End If
                        S(Serie) = Spiel
                        C(Serie) = 0
                    End If
                Next Serie
            RS_Spiele.MoveNext
        Loop
        RS_Spiele.Close
        For Serie = 0 To 4
            'Debug.Print "", SMax(Serie), CMax(Serie), Text(Serie)
            Debug.Print "", Text(Serie), CStr(CMax(Serie)) + _
                " Spiele beginnend bei Spiel " + CStr(SMax(Serie))
            Select Case j
            Case 0
                Sql = " (Heim='H') AND "
            Case 1
                Sql = " (Heim='A') AND "
            Case 2
                Sql = ""
            End Select
            Sql = "SELECT Gegner, Heim, ErgebnisA, Datum, Teamchef, Spiel " & _
                "FROM Spiel " & _
                "WHERE " + Sql + " (Spiel>= " + CStr(SMax(Serie)) + ") " & _
                "ORDER BY Spiel"
            Set RS_Spiele = CurrentDb.OpenRecordset(Sql)
            Dim Count As Integer
            Count = 0
            Do While Not RS_Spiele.EOF
                Debug.Print "", "", RS_Spiele("ErgebnisA") + " " + _
                    RS_Spiele("Heim") + " " + CStr(RS_Spiele("Datum")) + _
                    " " + CStr(RS_Spiele("Spiel")) + " " + _
                    RS_Spiele("Gegner"), RS_Spiele("Teamchef")
                Count = Count + 1
                If Count = CMax(Serie) Then
                    Exit Do
                End If
            RS_Spiele.MoveNext
        Loop
        RS_Spiele.Close
    Next Serie
Next j
End Sub
```